Chapter 16 Line and Surface Integrals

Useful Tip: If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then it is possible to view 3-D plots generated by *Mathematica* from different perspectives. First, place your screen cursor over the plot. Then drag the mouse while pressing down on the left mouse button to rotate the plot.

■ 16.1 Vector Fields

Students should read Section 16.1 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Let F_1, F_2, and F_3 be functions of x, y, and z. The vector-valued function

$$\mathbf{F}(x, y, z) = \langle F_1(x, y, z), F_2(x, y, z), F_3(x, y, z)\rangle$$

is called a *vector field*. We have already encountered a vector field in the form of the gradient of a function. Other useful examples of vector fields are the gravitational force, the velocity of fluid, magnetic fields, and electric fields.

We use the *Mathematica* commands **VectorFieldPlot** and **VectorFieldPlot3D** to plot the graphs of vector fields. However, before using these commands, it is advisable to load the **VectorFieldPlots** package. This is done by evaluating

In[505]:=
```
Needs["VectorFieldPlots`"]
```

Example 16.1. Draw the following vector fields.
a) $\mathbf{F}(x, y) = \langle \sin y, \cos x\rangle$ b) $\mathbf{F}(x, y, z) = \langle y, x + z, 2x - y\rangle$

Solution:
a)

In[506]:=
```
Clear[F, x, y, z]
F[x_, y_] = {Sin[y], Cos[x]}
```

Out[507]=
```
{Sin[y], Cos[x]}
```

In[508]:=
```
VectorFieldPlot[F[x, y], {x, -5, 5}, {y, -4, 4}, ImageSize → {250}]
```

Out[508]=



Here is another display of the preceding vector field with some options specified.

In[509]:=
```
VectorFieldPlot[F[x, y], {x, -5, 5}, {y, -4, 4}, Axes → True,
 AxesOrigin → {0, 0}, Frame → False, ColorFunction → Hue, ImageSize → {250}]
```

Out[509]=



To see other available options of **VectorFieldPlot**, evaluate the command **Options[VectorFieldPlot]**.

b) We shall use two of the options of **VectorFieldPlot3D**, which does not have as many options as **VectorFieldPlot**. (Again, you can find these by evaluating **Options[VectorFieldPlot3D**].)

In[510]:=
```
Clear[F, x, y, z]
F[x_, y_, z_] = {y z^2, x z^2, 2 x y z}
VectorFieldPlot3D[F[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
 ColorFunction → Hue, VectorHeads → True, ImageSize → {250}]
```

Out[511]= $\{y z^2, x z^2, 2 x y z\}$

Out[512]=



Example 16.2. Draw the unit radial vector fields:

a) $\mathbf{F}(x, y) = \left\langle \frac{x}{\sqrt{x^2+y^2}}, \frac{y}{\sqrt{x^2+y^2}} \right\rangle$ b) $\mathbf{F}(x, y, z) = \left\langle \frac{x}{\sqrt{x^2+y^2+z^2}}, \frac{y}{\sqrt{x^2+y^2+z^2}}, \frac{z}{\sqrt{x^2+y^2+z^2}} \right\rangle$

Solution: For convenience, we define both vector fields to be 0 at the origin. We shall use the **If** command to do so.

a)

In[513]:=
```
Clear[F, x, y]
F[x_, y_] = If[x^2 + y^2 ≠ 0, {x, y}/Sqrt[x^2 + y^2], {0, 0}]
VectorFieldPlot[F[x, y], {x, -3, 3}, {y, -3, 3}, ImageSize → {250}]
```

Out[514]= $\text{If}\left[x^2 + y^2 \neq 0, \frac{\{x, y\}}{\sqrt{x^2 + y^2}}, \{0, 0\}\right]$

Out[515]=



b)

In[516]:=
```
Clear[F, x, y, z]
F[x_, y_, z_] = If[x^2 + y^2 + z^2 ≠ 0, {x, y, z}/Sqrt[x^2 + y^2 + z^2], {0, 0, 0}]
VectorFieldPlot3D[F[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
 ColorFunction → Hue, VectorHeads → True, ImageSize → {250}]
```

Out[517]= $\text{If}\left[x^2+y^2+z^2\neq 0, \frac{\{x, y, z\}}{\sqrt{x^2+y^2+z^2}}, \{0, 0, 0\}\right]$

Out[518]=



Exercises

In Exercise 1 through 4, draw the given vector fields.

1. $\mathbf{F}(x, y) = \left\langle y^2 - 2xy, \; xy + 6x^2 \right\rangle$
2. $\mathbf{F}(x, y, z) = \langle \sin x, \cos y, xz \rangle$
3. $\mathbf{F}(x, y) = \left\langle -\frac{y}{\sqrt{x^2+y^2}}, \frac{x}{\sqrt{x^2+y^2}} \right\rangle$
4, $\mathbf{F}(x, y, z) = \langle x + \cos(xz), \; y \sin(xy), \; xz\cos(yz) \rangle$

In Exerices 5 and 6, calculate and plot the gradient vector field for each of the following functions.

5. $f(x, y) = \ln\left(x + y^2\right)$
6. $f(x, y, z) = \sin x \; (\cos z / y)$

16.2 Line Integrals

Students should read Section 16.2 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Suppose C is a smooth curve in space whose parametric equations are given by

$$x = x(t), \; y = y(t), \; z = z(t)$$

where $a \le t \le b$. Let $C_1, C_2, C_3, \ldots, C_N$ be a partition of the curve C with arc length $\Delta s_1, \Delta s_2, \Delta s_3, \ldots, \Delta s_N$ and let $P_1, P_2, P_3, \ldots, P_N$ be points on the subarcs.

If $f(x, y, z)$ is a function that is continuous on the curve C, then the *line integral of* f is defined by

$$\int_C f(x, y, z)\, ds = \lim_{\Delta s_i \to 0} \sum_{i=1}^{N} f(P_i)\, \Delta s_i$$

NOTE: If $\mathbf{c}(t) = \langle x(t), y(t), z(t) \rangle$ is the vector equation of the curve C, then it can be shown (refer to your calculus textbook) that

$$\int_C f(x, y, z)\, ds = \int_a^b f(\mathbf{c}(t))\, \|\mathbf{c}'(t)\|\, dt$$

In addition, if $\mathbf{F}(x, y, z) = \langle F_1, F_2, F_3 \rangle$ is a vector field that is continuous on C, then the *line integral* of $\mathbf{F}$ over C is given by

$$\int_C \mathbf{F}(x, y, z) \cdot d\mathbf{s} = \int_C (\mathbf{F} \cdot \mathbf{T})\, ds = \int_a^b \mathbf{F}(\mathbf{c}(t)) \cdot \mathbf{c}'(t)\, dt$$

where $\mathbf{T}$ is the unit vector $\mathbf{T} = \frac{\mathbf{c}'(t)}{\|\mathbf{c}'(t)\|}$ and $\mathbf{F} \cdot \mathbf{T}$ is the dot product of $\mathbf{F}$ and $\mathbf{T}$.

Example 16.3. Find $\int_C f(x, y, z)\, ds$, where $f(x, y, z) = x\,y + z^2$ and C is given by $x = t$, $y = t^2$, and $z = t^3$, for $0 \le t \le 1$.

Solution:

```
In[519]:= Clear[x, y, z, t, f, c]
          f[x_, y_, z_] = x^2 y + x z
          x[t_] = t
          y[t_] = t^2
          z[t_] = t^3
          c[t_] = {x[t], y[t], z[t]}
Out[520]= x^2 y + x z
Out[521]= t
Out[522]= t^2
Out[523]= t^3
Out[524]= {t, t^2, t^3}
```

In[525]:= $\int_0^1 f[x[t], y[t], z[t]]\, Norm[c'[t]]\, dt$

Out[525]= $-\frac{1}{76545\sqrt{\frac{7}{2}\left(2i+\sqrt{5}\right)}} 2(-1)^{1/4}\left(84987(-1)^{3/4}\sqrt{2i+\sqrt{5}} - 532\, i\sqrt{14}\, \text{EllipticE}\left[\text{ArcSin}\left[\frac{3+3i}{\sqrt{2\left(-2i+\sqrt{5}\right)}}\right], \frac{2i-\sqrt{5}}{2i+\sqrt{5}}\right] - 266\sqrt{70}\, \text{EllipticE}\left[\text{ArcSin}\left[\frac{3+3i}{\sqrt{2\left(-2i+\sqrt{5}\right)}}\right], \frac{2i-\sqrt{5}}{2i+\sqrt{5}}\right] + 415\, i\sqrt{14}\, \text{EllipticF}\left[\text{ArcSin}\left[\frac{3+3i}{\sqrt{2\left(-2i+\sqrt{5}\right)}}\right], \frac{2i-\sqrt{5}}{2i+\sqrt{5}}\right] + 266\sqrt{70}\, \text{EllipticF}\left[\text{ArcSin}\left[\frac{3+3i}{\sqrt{2\left(-2i+\sqrt{5}\right)}}\right], \frac{2i-\sqrt{5}}{2i+\sqrt{5}}\right]\right)$

Here is a numerical approximation of the preceding line integral.

In[526]:= `NIntegrate[f[x[t], y[t], z[t]] Norm[c'[t]], {t, 0, 1}]`

Out[526]= 1.16521

Example 16.4. Find $\int_C \mathbf{F}(x, y, z)\cdot d\mathbf{s}$, where $\mathbf{F}(x, y, z) = \langle x\, z,\ z\, y^2,\ y\, x^2 \rangle$ and the curve C is given by $x = 2\,t$, $y = \sin t$, and $z = \cos t$, $0 \le t \le 2\pi$.

Solution:

```
In[527]:= Clear[x, y, z, t, f, c]
F[x_, y_, z_] = {x z, z y^2, y x^2}
x[t_] = 2 t
y[t_] = Sin[t]
z[t_] = Cos[t]
c[t_] = {x[t], y[t], z[t]}
Out[528]= {x z, y^2 z, x^2 y}
Out[529]= 2 t
Out[530]= Sin[t]
Out[531]= Cos[t]
Out[532]= {2 t, Sin[t], Cos[t]}
In[533]:= ∫_0^(2 Pi) F[x[t], y[t], z[t]].c'[t] ⅆt
```

Out[533]= $\frac{9\pi}{4} - \frac{16\pi^3}{3}$

```
In[534]:= N[%]
Out[534]= -158.298
```

Exercises

1. Find $\int_C f(x, y, z)\, ds$, where:

a. $f(x, y, z) = x y^2 - 4 zy$ and C is given by $x = 2t$, $y = t^{2/3}$, and $z = 1 - 3t^2$, for $0 \le t \le 1$.

b. $f(x, y, z) = \frac{yz}{x}$ and C is given by $x = \ln t$, $y = t^2$, and $z = 3t$, for $3 \le t \le 5$.

2. Find $\int_C \mathbf{F}(x, y) \cdot d\mathbf{s}$, where:

a. $\mathbf{F}(x, y) = \langle e^{3x-2y}, e^{2x+3y} \rangle$ and C is given by $x = 2t$, $y = \sin t$, $0 \le t \le \pi$

b. $\mathbf{F}(x, y) = \langle x^2, yx + y^2 \rangle$ and C is the unit circle center at the origin.

3. Find $\int_C \mathbf{F}(x, y, z) \cdot d\mathbf{s}$, where:

a. $\mathbf{F}(x, y, z) = \langle xyz, -xz, xy \rangle$ and C is given by $x = t$, $y = 2t^2$, $z = 3t$ $0 \le t \le 1$

b. $\mathbf{F}(x, y, z) = \langle xy^3, z + x^2, z^3 \rangle$ and C is the line segment joining $(-1, 2, -1)$ and $(1, 3, 4)$.

16.3 Conservative Vector Fields

Students should read Section 16.3 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Let $\mathbf{F}(x, y, z) = \langle F_1, F_2, F_3 \rangle$ be a vector field. Let C_1 and C_2 be any two different curves with the same initial point P and end point Q. We say that the vector field $\mathbf{F}$ is *path independent* if

$$\int_{C1} \mathbf{F}(x, y, z) \cdot d\mathbf{s} = \int_{C2} \mathbf{F}(x, y, z) \cdot d\mathbf{s}$$

A vector field that is path independent is called *conservative.*

NOTE 1: A vector field **F** is conservative if

$$\int_C \mathbf{F}(x,\ y,\ z)\cdot d\mathbf{s} = 0$$

for every closed curve C.

NOTE 2: If $\mathbf{F} = \nabla u$ is the gradient of a function $u = u(x,\ y,\ z)$, then we say that u is the *potential* of **F**. Moreover, if the end points of C are P and Q, we have

$$\int_C \mathbf{F}(x,\ y,\ z)\cdot d\mathbf{s} = u(P) - u(Q)$$

In particular, if the curve is closed, that is, if $P = Q$, then

$$\int_C \mathbf{F}(x,\ y,\ z)\cdot d\mathbf{s} = 0$$

Therefore, gradient is conservative. The converse of this statement is true if its domain is an open connected domain.

NOTE 3: Let $F = \langle F_1,\ F_2\rangle$. If $\mathbf{F} = \nabla u = \left\langle \frac{\partial u}{\partial x},\ \frac{\partial u}{\partial y}\right\rangle$, then $F_1 = \frac{\partial u}{\partial x}$ and $F_2 = \frac{\partial u}{\partial y}$. Taking the partial derivative of $F_1$with respect to y and that of F_2 with respect to x and using the fact that $\frac{\partial^2 u}{\partial x \partial y} = \frac{\partial^2 u}{\partial y \partial x}$, we see that F_1 and F_2 must satisfy

$$\frac{\partial F_1}{\partial y} = \frac{\partial F_2}{\partial x}$$

This equation is used to check if a vector field is conservative. In that case, we solve $F_1 = \frac{\partial u}{\partial x}$ for u by integrating with respect to x and then use the equation $F_2 = \frac{\partial u}{\partial y}$ to find the constant of integration. Here is an example.

Example 16.5. Show that the vector function $\mathbf{F} = \left\langle 3\,x^2 - 2\,xy\ + 2,\ 6\,y^2 - x^2 + 3\right\rangle$ is conservative and find its potential.

Solution: Here, $F_1 = x\,y^2$ and $F_2 = x^2\,y$. We now compare $\frac{\partial F_1}{\partial y}$ and $\frac{\partial F_2}{\partial x}$ to verify if **F** is conservative.

```
In[535]:= Clear[x, y, F1, F2]
          F1[x_, y_] = 3 x^2 - 2 x y + 2
          F2[x_, y_] = 6 y^2 - x^2 + 3

Out[536]= 2 + 3 x^2 - 2 x y

Out[537]= 3 - x^2 + 6 y^2

In[538]:= D[F1[x, y], y]
          D[F2[x, y], x]

Out[538]= -2 x

Out[539]= -2 x
```

Thus, the vector field is conservative. To find its potential u, we integrate $F_1 = \frac{\partial u}{\partial x}$ with respect to x to get

In[540]:=
```
Clear[h, u]
u = Integrate[F1[x, y], x] + h[y]
```

Out[541]= $2x + x^3 - x^2 y + h[y]$

Note that the addition of $h(y)$ is necessary because the constant of integration may depend on y. We now solve the equation $F_2 = \frac{\partial u}{\partial y}$ for $h'(y)$.

In[542]:=
```
Clear[sol]
sol = Solve[D[u, y] == F2[x, y], h'[y]]
```

Out[543]= $\{\{h'[y] \to 3\,(1 + 2y^2)\}\}$

This means that $h'(y) = 3(1 + 2y^2)$.

In[544]:=
```
Integrate[sol[[1, 1, 2]], y]
```

Out[544]= $3y + 2y^3$

Hence, $h(y) = 3y + 2y^2$ and so $u(x, y) = 2x + x^3 - x^2 y + 3y + 2y^3$ is the potential of **F**.

NOTE 4: Let $\mathbf{F} = \langle F_1, F_2, F_3 \rangle$. If $\mathbf{F} = \nabla u = \left\langle \frac{\partial u}{\partial x}, \frac{\partial u}{\partial y}, \frac{\partial u}{\partial z} \right\rangle$, then $F_1 = \frac{\partial u}{\partial x}$, $F_2 = \frac{\partial u}{\partial y}$ and $F_3 = \frac{\partial u}{\partial z}$. Taking the partial derivative of $F_1$with respect to y and that of F_2 with respect to x and using the fact that $\frac{\partial^2 u}{\partial x \partial y} = \frac{\partial^2 u}{\partial y \partial x}$, we see that F_1 and F_2 must satisfy

$$\frac{\partial F_1}{\partial y} = \frac{\partial F_2}{\partial x}$$

Taking the partial derivative of $F_1$with respect to z and that of F_3 with respect to x and using the fact that $\frac{\partial^2 u}{\partial x \partial z} = \frac{\partial^2 u}{\partial z \partial x}$, we see that F_1 and F_3 must satisfy

$$\frac{\partial F_1}{\partial z} = \frac{\partial F_3}{\partial x}$$

The preceding two equations can be used to check if a vector field is conservative. If this the case, we solve $F_1 = \frac{\partial u}{\partial x}$ for u by integrating with respect to x and then use $F_2 = \frac{\partial u}{\partial y}$ to find the constant of integration. We show this by the following example.

Example 16.6. Show that the vector function $\mathbf{F} = \langle yz + yz\cos(xy),\ xz + xz\cos(xy),\ xy + \sin(xy) \rangle$ is conservative and find its potential.

Solution: Here, $F_1 = yz + yz\cos(xy)$, $F_2 = xz + xz\cos(xy)$, and $F_3 = xy + \sin(xy)$.

In[545]:=
```
Clear[x, y, F1, F2, F3]
F1[x_, y_, z_] = y z + y z Cos[x y]
F2[x_, y_, z_] = x z + x z Cos[x y]
F3[x_, y_, z_] = x y + Sin[x y]
```

Out[546]= $y\,z + y\,z\,\mathrm{Cos}[x\,y]$

Out[547]= $x\,z + x\,z\,\mathrm{Cos}[x\,y]$

Out[548]= $x\,y + \mathrm{Sin}[x\,y]$

We now compare $\frac{\partial F_1}{\partial y}$ and $\frac{\partial F_2}{\partial x}$:

```
In[549]:= D[F1[x, y, z], y]
          D[F2[x, y, z], x]
Out[549]= z + z Cos[x y] - x y z Sin[x y]
Out[550]= z + z Cos[x y] - x y z Sin[x y]
```

Next, we compare $\frac{\partial F_1}{\partial z}$ and $\frac{\partial F_2}{\partial x}$:

```
In[551]:= D[F1[x, y, z], z]
          D[F3[x, y, z], x]
Out[551]= y + y Cos[x y]
Out[552]= y + y Cos[x y]
```

Thus, the vector field is conservative. To find its potential u, we integrate $F_1 = \frac{\partial u}{\partial x}$ with repsetct to x to get

```
In[553]:= Clear[u, h]
          u = Integrate[F1[x, y, z], x] + h[y, z]
Out[554]= x y z + h[y, z] + z Sin[x y]
```

Note that the addition of $h(y, z)$ is necessary because the constant of intgeration can depend on y and z. We now solve the equation $F_2 = \frac{\partial u}{\partial y}$ for $\frac{\partial h}{\partial y}$.

```
In[555]:= Clear[sol]
          sol = Solve[D[u, y] == F2[x, y, z], ∂y h[y, z]]
Out[556]= {{h^(1,0)[y, z] → 0}}
```

This means that $\frac{\partial h}{\partial y} = 0$ and hence h is a function of z only. Next, we solve the equation $F_3 = \frac{\partial u}{\partial z}$ for $\frac{\partial h}{\partial z}$.

```
In[557]:= Clear[sol2]
          sol2 = Solve[D[u, z] == F3[x, y, z], ∂z h[y, z]]
Out[558]= {{h^(0,1)[y, z] → 0}}
```

Hence, $\frac{\partial h}{\partial z} = 0$ and we can take $h = 0$. Therefore, $u = x\,y\,z + z \sin(x\,y)$ is the potential for the vector field **F**.

■ Exercises

1. Show that the vector field $\mathbf{F} = \langle y^3 - 3x^2 y,\ 3xy^2 - x^3 \rangle$ is conservative and find its potential.

2. Show that the vector field $\mathbf{F} = \left\langle y\,z + \frac{2xy}{z},\ xz + \frac{x^2}{z},\ xy - \frac{x^2 y}{z^2} \right\rangle$ is conservative and find its potential.

3. Determine whether the vector field $\mathbf{F} = \langle x^2,\ yx + e^z,\ y\,e^z \rangle$ is conservative. If it is, find its potential.

■ 16.4 Parametrized Surfaces and Surface Integrals

Students should read Section 16.4 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

A *parametrized surface* is a surface whose points are given in the form

$$G(u, v) = (x(u, v),\ \ y(u, v),\ \ z(u, v))$$

where u and v (called *parameters*) are independent variables used to describe a domain D (called the *parameter domain*).

The command for plotting parametrized surfaces is **ParametricPlot3D**. This command has been discussed in Section 14.1.2 of this text.

Example 16.7. Plot the parametrized surface defined by $G(u, v) = (\cos u \sin v,\ 4 \sin u \cos v,\ \cos v)$ over the domain $D = \{(u, v) \mid 0 \leq u \leq 2\pi,\ 0 \leq v \leq 2\pi\}$.

Solution:

In[559]:=
```
ParametricPlot3D[{ Cos[u] Sin[v], 4 Sin[u] Cos[v], Cos[v]},
 {u, 0, 2 Pi}, {v, 0, 2 Pi}, ImageSize → {250}]
```

Out[559]=



Example 16.8. Plot the parametrized surface defined by $G(u, v) = \left(u \cos v,\ u \sin v,\ 1 - u^2\right)$ over the domain $D = \{(u, v) \mid 0 \leq u \leq 1,\ 0 \leq v \leq 2\pi\}$.

Solution:

In[560]:=
```
ParametricPlot3D[{u Cos[v] , u Sin[v] , 1 - u^2}, {u, 0, 1},
 {v, 0, 2 Pi}, ColorFunction → "BlueGreenYellow", ImageSize → {250},
 ImagePadding → {{15, 15}, {15, 15}}]
```

Out[560]=



NOTE: On a parametrized surface $G(u, v) = (x(u, v),\ y(u, v),\ z(u, v))$, if we fix one of the variables, we get a curve on the surface. The plot following shows the curves corresponding to $u = 3/4$ (latitude) and $v = 5\pi/3$ (longitude).

```
In[561]:= Clear[plot1, plot2, plot3]
plot1 = ParametricPlot3D[{u Cos[v] , u Sin[v] , 1 - u^2},
   {u, 0, 1}, {v, 0, 2 Pi}, ColorFunction → "BlueGreenYellow"];
plot2 = ParametricPlot3D[{ 3 / 4 Cos[v] , 3 / 4 Sin[v] , 7 / 16},
   {v, 0, 2 Pi}, PlotStyle → {Thickness[0.01], Red}];
plot3 = ParametricPlot3D[{ u Cos[5 Pi / 3] , u Sin[5 Pi / 3] , 1 - u^2},
   {u, 0, 1}, PlotStyle → {Thickness[0.01], Blue}];
Show[plot1, plot2, plot3, PlotRange → All, ImageSize → {250},
 ImagePadding → {{15, 15}, {15, 15}}]
```

Out[565]=



Let $P = G(u_0, v_0)$ be a point on the parametrized surface S. For fixed $v = v_0$, the tangent vector to the curve $G(u, v_0)$ at (u_0, v_0) is given by

$$\mathbf{T}_u = \frac{\partial G}{\partial u}(u_0, v_0)$$

while the tangent vector for $G(u_0, v)$ corresponding to a fixed $u = u_0$ is given by

$$\mathbf{T}_v = \frac{\partial G}{\partial v}(u_0, v_0)$$

These two vectors are tangent to the surface S. Thus, the normal vector $\mathbf{n}$ to the tangent plane at $G(u_0, v_0)$ is given by

$$\mathbf{n}(P) = \mathbf{n}(u_0, v_0) = \mathbf{T}_u \times \mathbf{T}_v$$

Example 16.9. Consider the parametrized surface $G(u, v) = (u \cos v,\ u \sin v,\ 1 - v^2)$.

a) Find $\mathbf{T}_u$, $\mathbf{T}_v$, and $\mathbf{n}$.

b) Find the equation of the tangent plane at $(1/2,\ 5\pi/3)$.

c) Plot the tangent plane and surface.

Solution: Let us define G as a function of u and v in *Mathematica.*

```
In[566]:= Clear[G, u, v]
G[u_, v_] = {u Cos[v], u Sin[v], 1 - u^2}

Out[567]= {u Cos[v], u Sin[v], 1 - u^2}
```

a) We use **Tu** for $\mathbf{T}_u$ and **Tv** for $\mathbf{T}_v$. We evaluate these as functions of u and v.

```
In[568]:= Clear[Tu, Tv, n]
          Tu[u_, v_] = D[G[u, v], u]
          Tv[u_, v_] = D[G[u, v], v]
          n[u_, v_] = Cross[Tu[u, v], Tv[u, v]]
```

Out[569]= $\{\text{Cos}[v], \text{Sin}[v], -2u\}$

Out[570]= $\{-u\,\text{Sin}[v], u\,\text{Cos}[v], 0\}$

Out[571]= $\{2u^2\,\text{Cos}[v], 2u^2\,\text{Sin}[v], u\,\text{Cos}[v]^2 + u\,\text{Sin}[v]^2\}$

b) The normal vector to the tangent plane at $(1/2,\ 5\pi/3)$ is

```
In[572]:= Clear[normal]
          normal = n[1 / 2, 5 Pi / 3]
```

Out[573]= $\left\{\frac{1}{4}, -\frac{\sqrt{3}}{4}, \frac{1}{2}\right\}$

The tangent plane passes through the point

```
In[574]:= Clear[point]
          point = G[1 / 2, 5 Pi / 3]
```

Out[575]= $\left\{\frac{1}{4}, -\frac{\sqrt{3}}{4}, \frac{3}{4}\right\}$

Thus, the equation of the tangent plane is given by

```
In[576]:= Clear[tplane]
          tplane = normal.({x, y, z} - point) == 0
```

Out[577]= $\frac{1}{4}\left(-\frac{1}{4}+x\right)-\frac{1}{4}\sqrt{3}\left(\frac{\sqrt{3}}{4}+y\right)+\frac{1}{2}\left(-\frac{3}{4}+z\right)==0$

which simplifies to

```
In[578]:= Simplify[tplane]
```

Out[578]= $2x + 4z == 5 + 2\sqrt{3}\,y$

c) Here is the plot of the surface and the tangent plane. Observe that we have used **ColorFunction** and **ColorFunctionScaling** options.

In[579]:=
```
Clear[plot1, plot2]
plot1 = ParametricPlot3D[G[u, v],
   {u, 0, 1}, {v, 0, 2 Pi}, ColorFunction → "BlueGreenYellow"];
plot2 = ContourPlot3D[2 x + 4 z == 5 + 2 √3 y, {x, -3, 3}, {y, -3, 3},
   {z, -4, 4}, ColorFunction → Function[{x, y, z}, Hue[Mod[z, 1]]],
   ColorFunctionScaling → False];
Show[plot1, plot2, ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[582]=



NOTE: The area $A(S)$ of a parametrized surface S: $G(u, v) = (x(u, v), \ y(u, v), \ z(u, v))$, where $(u, v) \in D$, is given by

$$A(S) = \int\int_D \|\mathbf{n}(u, v)\| \, du \, dv$$

If $f(x, y, z)$ is continuous at all points of S, then the surface area of f over S is given by

$$\int\int_S f(x, y, z) \, dS = \int\int_D f(G(u, v)) \, \|\mathbf{n}(u, v)\| \, du \, dv$$

Example 16.10. Show the following:
a) The area of the cylinder of height h and radius r is $2\pi rh$.
b) The area of the sphere of radius r is $4\pi r^2$.

Solution:
a) A parametric equation of the cylinder of height h and radius r can be given by

$$x = r\cos v, \ y = r\sin v, \text{ and } z = u, \text{ where } \ 0 \le v \le 2\pi, \ 0 \le u \le h$$

Thus, the cylinder is given by $G(u, v) = (r\cos u, \ r\sin u, \ v)$.

In[583]:=
```
Clear[ G, u, v, r]
G[u_, v_] = {r Cos[v], r Sin[v], u}
```

Out[584]= {r Cos[v], r Sin[v], u}

Here is a plot of the cylinder with $r = 3$ and $h = 5$:

In[585]:=
```
r = 3; h = 5;
ParametricPlot3D[G[u, v], {u, 0, h}, {v, 0, 2 Pi}]
```

Out[586]=



To compute the surface area of the cylinder, we need to compute its normal vector.

In[587]:=
```
Clear[Tu, Tv, n, r, h]
Tu[u_, v_] = D[G[u, v], u];
Tv[u_, v_] = D[G[u, v], v];
n[u_, v_] = Cross[Tu[u, v], Tv[u, v]]
```

Out[590]= {- r Cos[v], - r Sin[v], 0}

Here is a plot of the cylinder with its normal vector for $r = 3$ and $h = 5$:

In[591]:=
```
r = 3; h = 5;
Clear[plot1, plot2]
plot1 = ParametricPlot3D[G[u, v], {u, 0, h}, {v, 0, 2 Pi}];
plot2 = VectorFieldPlot3D[n[u, v], {u, 0, h},
   {v, -2 Pi, 2 Pi}, {z, -3, 3},  VectorHeads → True, PlotPoints → 15];
Show[plot1, plot2, ImageSize → {250}]
Clear[r, h]
```

Out[595]=



The surface area is

In[597]:= SArea = $\int_0^h \int_0^{2\,Pi}$ Norm[n[u, v]] ⅆv ⅆu

Out[597]= 2 h π Abs[r]

Since $r > 0$, $|r| = r$ and hence the preceding output is $2\pi r h$.

b) A parametric equation of the sphere of radius r is

$$x = r\cos u \sin v,\ y = r\sin u \sin v,\ z = r\cos v$$

where $0 \le u \le 2\pi$ and $0 \le v \le \pi$. Thus, the sphere is given by $G(u, v) = (r\cos u \sin v,\ r\sin u \sin v,\ r\cos v)$.

In[598]:=
```
Clear[ G, u, v, r]
G[u_, v_] = {r Cos[u] Sin[v],  r Sin[u] Sin[v], r Cos[v]}
```

Out[599]= {r Cos[u] Sin[v], r Sin[u] Sin[v], r Cos[v]}

Here is a plot of the sphere with $r = 3$.

In[600]:=
```
r = 3;
ParametricPlot3D[G[u, v], {u, 0, 2 Pi}, {v, 0, Pi}, ImageSize → {250}]
```

Out[601]=



To compute the surface area of the sphere, we need to compute its normal vector.

In[602]:=
```
Clear[Tu, Tv, n, r]
Tu[u_, v_] = D[G[u, v], u];
Tv[u_, v_] = D[G[u, v], v];
n[u_, v_] = Cross[Tu[u, v], Tv[u, v]]
```

Out[605]=
```
{-r^2 Cos[u] Sin[v]^2, -r^2 Sin[u] Sin[v]^2,
 -r^2 Cos[u]^2 Cos[v] Sin[v] - r^2 Cos[v] Sin[u]^2 Sin[v]}
```

Here is a plot of the sphere with its normal vector for $r = 3$.

```
In[606]:= r = 3; h = 5;
          Clear[plot1, plot2]
          plot1 = ParametricPlot3D[G[u, v], {u, 0, 2 Pi}, {v, 0, h}];
          plot2 = VectorFieldPlot3D[n[u, v], {u, -2 Pi, 2 Pi},
             {v, 0, h}, {z, -3, 3}, VectorHeads → True, PlotPoints → 10];
          Show[plot1, plot2, ImageSize → {250}]
          Clear[r, h]
```

Out[610]=



The surface area is

In[612]:= $\texttt{SArea} = \int_0^{\texttt{Pi}} \int_0^{\texttt{2 Pi}} \texttt{Norm[n[u, v]]}\, du\, dv$

Out[612]= $4 \pi\, r\, \texttt{Conjugate}[r]$

For a real number r, the conjugate of r is r and hence the preceding output is $4 \pi r^2$.

Example 16.11. Consider the parametrized surface S defined by $G(u, v) = (u \cos v,\ u \sin v,\ v)$, where $0 \le u \le 1$, $0 \le v \le 2\pi$.
a) Find the surface area of S.
b) Evaluate $\iint_S xyz\, dS$.

Solution:

a)

```
In[613]:= Clear[ G, u, v]
          G[u_, v_] = {u Cos[v], u Sin[v], v}
```

Out[614]= {u Cos[v], u Sin[v], v}

```
In[615]:= Clear[Tu, Tv, n]
          Tu[u_, v_] = D[G[u, v], u]
          Tv[u_, v_] = D[G[u, v], v]
          n[u_, v_] = Cross[Tu[u, v], Tv[u, v]]

Out[616]= {Cos[v], Sin[v], 0}

Out[617]= {-u Sin[v], u Cos[v], 1}

Out[618]= {Sin[v], -Cos[v], u Cos[v]^2 + u Sin[v]^2}
```

The surface area $A(S)$ is given by

```
In[619]:= SArea = ∫_0^1 ∫_0^(2 Pi) Norm[n[u, v]] dv du

Out[619]= π (√2 + ArcSinh[1])
```

which is approximately equal to

```
In[620]:= N[%]

Out[620]= 7.2118
```

b) We define f:

```
In[621]:= Clear[f]
          f[x_, y_, z_] = x y z

Out[622]= x y z
```

The surface integral of f is

```
In[623]:= ∫_0^1 ∫_0^(2 Pi) f[G[u, v][[1]], G[u, v][[2]], G[u, v][[3]]] Norm[n[u, v]] dv du

Out[623]= -1/16 π (3 √2 - ArcSinh[1])
```

Or numerically,

```
In[624]:= N[%]

Out[624]= -0.659983
```

Exercises

1. Plot the parametrized surface $G(u, v) = (e^u \sin v, e^u \cos v, v)$ over the domain $D = \{(u, v) \mid -1 \le u \le 1, \ 0 \le v \le 2\pi\}$.

2. Plot the parametrized surface $G(u, v) = (3 \sin u \cos v, \sin u \sin v, \cos v + 3 \cos u)$ over the domain $D = \{(u, v) \mid 0 \le u \le 2\pi, \ 0 \le v \le 2\pi\}$.

3. Consider the parametrized surface $G(u, v) = (e^{-u} \cos v, e^u \sin v, e^u \cos v)$.
a. Find $\mathbf{T}_u$, $\mathbf{T}_v$, and $\mathbf{n}$.
b. Find the equation of the tangent plane at $(0, \pi/2)$.
c. Plot the tangent plane and surface.

4. Consider the parametrized surface S: $G(u, v) = \left(u - v,\ 3u + v,\ u^2 - 2uv + 6v^2\right)$, where $0 \le u \le 1$, $0 \le v \le 1$.

a. Find the surface area of S. (Use **NIntegrate** for faster integration.)

b. Evaluate $\int\int_S \left(3x + 2y^2 - z^2\right) dS$.

■ 16.5 Surface Integrals of Vector Fields

Students should read Section 16.5 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

An orientation of a surface S is a continuously varying choice of the unit normal vector $\mathbf{e}_n(P)$ at each point of the surface. Thus, $\mathbf{e}_n$ is given by either

$$\mathbf{e}_n(P) = \frac{\mathbf{n}(P)}{||\mathbf{n}(P)||} \quad \text{or} \quad \mathbf{e}_n(P) = -\frac{\mathbf{n}(P)}{||\mathbf{n}(P)||}$$

If $\mathbf{F}(x, y, z)$ is continuous at all points of a parametrized surface S, then the *surface integral* of $\mathbf{F}$ over S is given by

$$\int\int_S \mathbf{F} \cdot d\mathbf{S} = \int\int_S (\mathbf{F} \cdot \mathbf{e}_n)\, dS$$

where $\mathbf{e}_n$ is the unit normal determined by an orientation. The surface integral of $\mathbf{F}$ is also called the *flux* of $\mathbf{F}$ across S.

The surface integral of $\mathbf{F}$ over a parametrized surface S given by $G(u, v) = (x(u, v),\ y(u, v),\ z(u, v))$, where $(u, v) \in D$, is given by

$$\int\int_S \mathbf{F} \cdot d\mathbf{S} = \int\int_S (\mathbf{F} \cdot \mathbf{e}_n)\, dS = \int\int_D \mathbf{F}(G(u, v)) \cdot \mathbf{n}(u, v)\, du\, dv$$

Example 16.12. Find $\int\int_S \mathbf{F} \cdot d\mathbf{S}$, where $\mathbf{F}(x, y, z) = \langle x z,\ z,\ y x \rangle$ and S is given by $G(u, v) = \left(u - v^2,\ u v,\ u^2 - v\right)$, $0 \le u \le 2$, and $1 \le v \le 3$.

Solution:

```
In[625]:= Clear[F, G, x, y, z, u, v]
          F[x_, y_, z_] = {x z, z, y x}
          G[u_, v_] = {u - v^2, u v, u^2 - v}

Out[626]= {x z, z, x y}

Out[627]= {u - v^2, u v, u^2 - v}

In[628]:= Clear[Tu, Tv, n]
          Tu[u_, v_] = D[G[u, v], u]
          Tv[u_, v_] = D[G[u, v], v]
          n[u_, v_] = Cross[Tu[u, v], Tv[u, v]]

Out[629]= {1, v, 2 u}

Out[630]= {-2 v, u, -1}

Out[631]= {-2 u^2 - v, 1 - 4 u v, u + 2 v^2}

In[632]:= Flux = ∫_0^2 ∫_1^3 F[G[u, v][[1]], G[u, v][[2]], G[u, v][[3]]] .n[u, v] dv du

Out[632]= -6928/15
```

Example 16.13. Find $\iint_S \mathbf{F}\cdot d\mathbf{S}$, where $\mathbf{F}(x, y, z) = \langle x^2, z^2, y + x^2\rangle$ and S is the upper hemisphere $x^2 + y^2 + z^2 = 4$ with outward normal orientation.

Solution: First, we find the parametric equation of the cylinder. This can be given by $x = 2\cos u \sin v$, $y = 2\sin u \sin v$, and $z = 2\cos v$, where $0 \le u \le 2\pi$ and $0 \le v \le \pi/2$.

For the hemisphere to have the outward orientation, we note that $\mathbf{n} = \mathbf{T}_v \times \mathbf{T}_u$. With this in mind we compute the flux of $\mathbf{F}$ across S through the following steps.

```
In[633]:= Clear[F, G, x, y, z, u, v]
          F[x_, y_, z_] = { x^2, z^2, x^2 + y + z^3}
          G[u_, v_] = { 2 Cos[u] Sin[v] , 2 Sin[u] Sin[v] , Cos[v] }

Out[634]= {x^2, z^2, x^2 + y + z^3}

Out[635]= {2 Cos[u] Sin[v], 2 Sin[u] Sin[v], Cos[v]}

In[636]:= Clear[Tu, Tv, n]
          Tu[u_, v_] = D[G[u, v], u]
          Tv[u_, v_] = D[G[u, v], v]
          n[u_, v_] = Cross[Tv[u, v], Tu[u, v]]

Out[637]= {-2 Sin[u] Sin[v], 2 Cos[u] Sin[v], 0}

Out[638]= {2 Cos[u] Cos[v], 2 Cos[v] Sin[u], -Sin[v]}

Out[639]= {2 Cos[u] Sin[v]^2, 2 Sin[u] Sin[v]^2, 4 Cos[u]^2 Cos[v] Sin[v] + 4 Cos[v] Sin[u]^2 Sin[v]}
```

In[640]:= $\text{Flux} = \int_0^{\text{Pi}/2}\int_0^{2\,\text{Pi}} \text{F[G[u, v][[1]], G[u, v][[2]], G[u, v][[3]]] . n[u, v]}\, du\, dv$

Out[640]= $\frac{28\pi}{5}$

Exercises

1. Find $\iint_S \mathbf{F}\cdot d\mathbf{S}$, where $\mathbf{F}(x, y, z) = \langle e^z, z, yx\rangle$ and S is given by $G(u, v) = (uv, u - v, u)$, $0 \le u \le 2$, and $-1 \le v \le 1$, and oriented by $\mathbf{n} = \mathbf{T}_u \times \mathbf{T}_v$.

2. Find $\iint_S \mathbf{F}\cdot d\mathbf{S}$, where $\mathbf{F}(x, y, z) = \langle z, x, y\rangle$ and S is the portion of the ellipsoid $\frac{x^2}{16} + \frac{y^2}{9} + \frac{z^2}{4} = 1$ for which $x \le 0$, $y \le 0$, and $z \le 0$ with outward normal orientation.

3. Let S be given by $G(u, v) = \left(\left(1 + v\cos\frac{u}{2}\right)\cos u, \left(1 + v\cos\frac{u}{2}\right)\sin u, v\sin\frac{u}{2}\right)$, $0 \le u \le 2\pi$, and $\frac{-1}{2} \le v \le \frac{1}{2}$.

a. Plot the surface S. (S is an example of a *Mobius strip*.)

b. Find the surface area of S.

c. Evaluate $\iint_S (x^2 + 2y^2 + 3z^2)\, dS$.

d. Find the intersection points of S and the xy-plane.

e. For each of the points on the intersection of S and the xy-plane, find the normal vector $\mathbf{n}$.

f. Show that $\mathbf{n}$ varies continuously but that $\mathbf{n}(2\pi, 0) = -\mathbf{n}(u, 0)$. (This shows that S is not orientable and hence it is impossible to integrate a vector field over S.)